Calculus WIZ и *The Mathematical Explorer* – напреднало ниво

Числени и символни възможности

И двата продукта *Calculus WIZ* и *The Mathematical Explorer* поддържат много символни и изчислителни възможности от техния изходен продукт *Mathematica.*

Можете да използвате произволната точност на изчислителната аритметика на *Mathematica*, за да пресмятате прочутите математически константи с колкото десетични знаци пожелаете:

```
N[Pi, 120]

3.14159265358979323846264338327950288419716939937510582097494459230781640628620899862803482534211706798214808651328230665
```

Тук даваме числото на Лудолф и константата на Ойлер със 120 десетични знака:

```
N[E, 120]

2.71828182845904523536028747135266249775724709369995957496696762772407663035354759457138217852516642742746639193200305992
```

Също така, можете да работите и символно с отношенията между константите:

```
E^(I Pi)

-1
```

Понякога не може да получите веднага отговора, както в долния пример за изчисляване на имагинерната единица, повдигната на степен самата себе си:

```
I^I
```

i^{i}

Числената стойност на този израз е:

```
N[I^I]
```

0.20788 + 0. i

The Mathematical Explorer може да даде лесно и символната му стойност:

```
ComplexExpand[I^I]
```

$e^{-\pi/2}$

Впрочем, тази команда на *Mathematica* не е включена в *Calculus WIZ*. Ако наберете

```
ComplexExpand[I^I]
```

в сесия на *Calculus WIZ*, отговорът ще е следният:

Disabled`ComplexExpand[i^i]

Лесно е да намерите решението на алгебрично уравнение

```
Solve[x^3 == 1, x]
```

$\{\{x \to 1\}, \{x \to -\sqrt[3]{-1}\}, \{x \to (-1)^{2/3}\}\}$

и да използвате отговора в следващи изчисления. Имайте предвид, че символът % съответства на резултата от предишното изчисление, а символът /. извършва субституция, т.е. x е заместено със стойността от дясната страна на стрелката (x /. x->a дава a) :

```
x /. %
```

$\{1, -\sqrt[3]{-1}, (-1)^{2/3}\}$

Получените комплексни числа, могат да се преобразуват в тяхната алгебрична форма:

```
ComplexExpand[%]
```

$\left\{1, -\frac{1}{2} - \frac{i\sqrt{3}}{2}, -\frac{1}{2} + \frac{i\sqrt{3}}{2}\right\}$

и/или да се пресметнат числено:

```
N[%]
```

$\{1., -0.5 - 0.866025\, i, -0.5 + 0.866025\, i\}$

Като пример за символно изчисление за напреднали, нека се търси преобразуването на тригонометрична и хиперболична функция в експоненциална функция и обратно.

```
TrigToExp[Sin[z]]
```

$$\frac{1}{2}\, i\, e^{-i z} - \frac{1}{2}\, i\, e^{i z}$$

Това трансформира комплексните експоненти към тригонометрична и хиперболична функции:

```
ExpToTrig[%]
```

Sin[z]

Резултатът не ни учудва. Но следващият пример ще ни учуди. Нека преобразуваме синус на трета степен в експоненциална функция, да развием степента с биномната формула и да преобразуваме резултата обратно в тригонометрична функция. Ето какво получаваме:

```
TrigToExp[Sin[z]^3]
```

$$-\frac{1}{8}\, i \left(e^{-i z} - e^{i z}\right)^3$$

```
Expand[%]
```

$$\frac{3}{8}\, i\, e^{-i z} - \frac{3}{8}\, i\, e^{i z} - \frac{1}{8}\, i\, e^{-3 i z} + \frac{1}{8}\, i\, e^{3 i z}$$

```
ExpToTrig[%]
```

$$\frac{3\,\mathrm{Sin}[z]}{4} - \frac{1}{4}\,\mathrm{Sin}[3 z]$$

Ако се съмнявате, че

$$\frac{3\,\mathrm{Sin}[z]}{4} - \frac{1}{4}\,\mathrm{Sin}[3 z] == \mathrm{Sin}[z]\text{^}3$$

прегледайте някои математически таблици или просто напишете

$$\texttt{Simplify}\left[\frac{3\,\texttt{Sin}[z]}{4} - \frac{1}{4}\,\texttt{Sin}[3\,z] == \texttt{Sin}[z]\,\texttt{^}3\right]$$

Ще видите, че това е

True

Фукнцията ExpToTrig се поддържа и в двата продукта и може да се използва вместо функцията ComplexExpand, когато се работи с изрази като $-\sqrt[3]{-1}$:

$$\texttt{ExpToTrig}\left[-(-1)^{1/3}\right]$$

$$-\frac{1}{2} - \frac{i\sqrt{3}}{2}$$

Програмни възможности на Calculus WIZ и The Mathematical Explorer

Като продукти, основани на *Mathematica*, *Calculus WIZ* и *The Mathematical Explorer* поддържат процедурно, функционално програмиране (LISP програмен стил, използващ чисти функции) и програмни парадигми, базирани на правила. По тази причина, техните възможности не са ограничени с предефинирани темплейти и те могат да се използват по конструктивен начин за писане на потребителски програми.

Лесно можем да генерираме таблици със стойности на функции:

```
Table[Sin[x], {x, 0, Pi, Pi/6}]
```

$$\left\{0,\ \frac{1}{2},\ \frac{\sqrt{3}}{2},\ 1,\ \frac{\sqrt{3}}{2},\ \frac{1}{2},\ 0\right\}$$

Като се възползваме от функционалния програмен стил, можем да пишем кратки елегантни програми. Тук създаваме таблица от производните и интегралите на даден списък от функции:

```
({#, D[#, x], Integrate[#, x]} & /@ {Sin[x], Cos[x], Tan[x], Cot[x], ArcTan[x]}) // TableForm
```

Sin[x]	Cos[x]	-Cos[x]
Cos[x]	-Sin[x]	Sin[x]
Tan[x]	$\text{Sec}[x]^2$	-Log[Cos[x]]
Cot[x]	$-\text{Csc}[x]^2$	Log[Sin[x]]
ArcTan[x]	$\frac{1}{1+x^2}$	$x\,\text{ArcTan}[x] - \frac{1}{2}\text{Log}\left[1+x^2\right]$

Също така лесно се създават графики на функции и комбинации от тях:

```
Plot[{Sin[x], Cos[x]}, {x, 0, 2 Pi}]
```

Програмирането, базирано на правила може да се използва за извършване на полагане (субституция) при решаване на уравнения:

```
Sin[x]^2+Cos[x]==1/2  /. Sin[x] -> Sqrt[1-Cos[x]^2]
```

$$1 + \mathrm{Cos}[x] - \mathrm{Cos}[x]^2 == \frac{1}{2}$$

Символът % се отнася за резултата от предишното изчисление:

```
% /. Cos[x] -> z
```

$$1 + z - z^2 == \frac{1}{2}$$

Командата Solve решава полученото уравнение, което заместваме с %:

```
Solve[%, z]
```

$$\left\{\left\{z \to \frac{1}{2}\left(1 - \sqrt{3}\right)\right\}, \left\{z \to \frac{1}{2}\left(1 + \sqrt{3}\right)\right\}\right\}$$

Резултатите могат веднага да се използват за обратно заместване:

```
Cos[x] == z /. %
```

$$\left\{\mathrm{Cos}[x] == \frac{1}{2}\left(1 - \sqrt{3}\right), \mathrm{Cos}[x] == \frac{1}{2}\left(1 + \sqrt{3}\right)\right\}$$

и можем да решим например първото от тези две гонометрични уравнения:

```
Solve[%[[1]], x]
```

Solve::ifun Inverse functions are

being used by Solve, so some solutions

may not be found.

$$\left\{\left\{x \to -\mathrm{ArcCos}\left[\frac{1}{2}\left(1 - \sqrt{3}\right)\right]\right\}, \left\{x \to \mathrm{ArcCos}\left[\frac{1}{2}\left(1 - \sqrt{3}\right)\right]\right\}\right\}$$

Забележете, че предупредителното съобщение беше генерирано, тъй като функцията не е обратима върху цялата област. Тъй като и двата продукта *Calculus WIZ* и *The Mathematical Explorer* поддържат изчисления с точни величини, предишният изход не е записан като число. Впрочем, ние можем, да поискаме числените значения на решенията:

```
N[%]
```

$$\{\{x \to -1.94553\}, \{x \to 1.94553\}\}$$

Графични възможности

Двата продукта поддържат голям брой от графични преставяния на обекти: основни графични модели, (т.н. графични примитиви), 2D и 3D графики на функции и контурни графики.

Следващата команда ще начертае точка и права.

```
Show[Graphics[{Line[{{0, 0}, {1, 1}}], PointSize[0.02],
   Point[{1/2, 1}]}]]
```



Задача за изследване: Намерете пресечните точки на графиките на функциите Sin(x) и x^2 в първи квадрант.

За да имаме представа как изглеждат нещата, начертаваме графиките на двете функции:

```
Plot[{Sin[x], x^2}, {x, 0, Pi}]
```



Сега можем да търсим пресечната точка:

```
Solve[Sin[x] == x^2, x]
```

Solve::tdep : The equations appear to involve the variables to be solved for in an essentially non-algebraic way.

$\text{Solve}[\sin(x) == x^2, x]$

В общия случай, не е възможно да се намери решението на трансцендентно уравнение в затворен вид. Затова сме ограничени да търсим само числено решение с приближен числен метод. За целта прилагаме функцията FindRoot:

```
{x, Sin[x]} /. FindRoot[Sin[x] == x^2, {x, 0.1}]
```

$\{-2.68703 \times 10^{-8}, -2.68703 \times 10^{-8}\}$

Хванати сме в случая от тривиалното решение (стойността 0.1 е начална стойност на x при стартиране на търсенето). От графиката по-горе, виждаме, че второто решение е някъде около 0.5 и 1. Нека започнем с начална стойност 0.5:

```
pt = {x, Sin[x]} /. FindRoot[Sin[x] == x^2, {x, 0.5}]
```

{0.876726, 0.768649}

Този път хванахме нетривиалното решение. Можем да го изобразим и на фигурата:

```
Plot[{Sin[x], x^2}, {x, 0, Pi},
 Epilog -> {PointSize[0.03], Point[{0, 0}], Point[pt]}]
```



Обработка на данни

Поради факта, че Calculus WIZ и The Mathematical Explorer поддържат основните обработки на данни, тяхната употреба не е ограничеа само до класовете по математика.

Тук се генерират няколко случайни числа (в реалността, тези данни ще дойдат от измервания):

```
data = Table[{k, k^2 + 2 k + Random[] / 10}, {k, 1, 10}]
```

Ще отбележим, че Calculus WIZ няма ккоманда Random[], така че имаме нужда от някои външни реални данни, за да извършим следващото изчисление.

$$\begin{pmatrix} 1 & 3.07191 \\ 2 & 8.0146 \\ 3 & 15.0765 \\ 4 & 24.034 \\ 5 & 35.0254 \\ 6 & 48.0838 \\ 7 & 63.0064 \\ 8 & 80.0063 \\ 9 & 99.0093 \\ 10 & 120.059 \end{pmatrix}$$

```
ListPlot[data, PlotStyle→ {PointSize[0.03]},
 AxesOrigin→ {0, 0}]
```



Тази команда намира приближение към списъка от данни по метода на най-малките квадрати с функциите 1, x и x^2.

```
Fit[data, {1, x, x^2}, x]
```

$$1.00074x^2 + 1.9887x + 0.0725026$$

Нека да изобразим данните и тяхното приближение на обща графика:

```
Plot[1.0007380890493205` x^2 + 1.988699387191992` x +
  0.07250258820562902`, {x, 0, 10},
 Epilog :> {PointSize[0.02], Point /@ data}]
```



Накрая ще отбележим, че Calculus WIZ и The Mathematical Explorer не могат да импортират и експортират данни във файлове. Данните трябва да се прехвърлят със системния буфер Clipboard.

Заключение

Програмите *Calculus WIZ* и *The Mathematical Explorer* са самостоятелни продукти, базирани на стабилната технология на *Mathematica*. Те са подходящи не само за изследване на математически идеи с готови моделни темплейти, но и с програми, написани от студентите. И двата продукта поддържат основния език за програмиране на *Mathematica* (с някои ограничения), с неговите базирани на правила функционални и процедурни парадигми. Двата продукта поддържат също така много команди на *Mathematica* за манипулиране на формули, символно и числено интегриране и др. Те не поддържат командите на *Mathematica* за обработка на големи масиви от данни, защото тези команди не се предполага, че ще се използват от студенти в горните класове на училищата или в първи курс на университета. Има и някои други ограничения спрямо пълната версия на *Mathematica*. *Calculus WIZ* и *The Mathematical Explorer* се препоръчват за късните курсове на гимназиите и началните курсове на университетите и колежите.